

## Scilab no Processamento Digital de Sinais – parte 2

### 4. Polinômios

Para definirmos um polinômio, utiliza-se a função *poly*.  
O polinômio pode ser especificado a partir dos seus coeficientes ou de suas raízes, como segue:

```
// Criando um polinômio a partir dos seus coeficientes \
q1 = poly ([2 -3 1], 'z', 'c')
```

```
// Calculando as raízes de um polinômio
```

```
roots(q1)
```

```
// Criando um polinômio a partir das suas raízes \
```

```
q2 = poly ([1 2], 'z')
```

Pode-se executar operações sobre os polinômios como qualquer outro objeto no Scilab (escalares, vetores e matrizes)

```
q3 = q1 + q2
```

```
q4 = q1 - q2
```

```
q5 = q1 * q2
```

```
q6 = q1 / q2
```

#### 4.1. Deteminação do valor do polinômio em um ponto :

Para calcular o valor de um polinômio em um ponto específico de seu domínio, utiliza-se a função *freq* :

```
pv=freq(num,den,v)
```

onde *pv* é a resposta, *num* é o polinômio numerador, *den* é o polinômio denominador e *v* é a faixa de valores nos quais deseja-se a avaliação do polinômio.

Ex :

*// determinando os coeficientes dos polinômios*

*h=[1:5,4:-1:1]*

*// montando o polinômio*

*hz=poly(h,'z','c')*

*// determinando as faixas de valores nas quais se deseja calcular o resultado*

*--> f=(0.:1:1)*

*// calculando*

*--> hf=freq(hz,1,exp(%pi\*i\*f))*

## 5. Funções de transferência

DSP (digital signal processing) faz uso de polinômios racionais para modelar sinais e sistema de transferência. A função que processa um polinômio racional de modo a representar uma função de transferência é denominada *syslin* :

*sl=syslin(domain,num,den)*

Ex :

*// criando um polinômio do tipo iir*


*h1=iir(3,'lp','butt',[.3 0],[0 0])*

*// convertendo este polinômio em um sistema linear*

*h1=syslin('d',h1);*

### 5.1 Interconectando sistemas (funções de transferência) :

Sistemas lineares criados no ambiente do Scilab podem ser conectados em cascata ou em paralelo. As quatro possibilidades de interconexão são mostrados abaixo, onde  $s_1$  e  $s_2$  são as funções de transferência que representam cada bloco do sistema:



## 6. Filtrando um sinal

A resposta temporal de um sistema linear pode ser determinada pela função *flts* :

```
y=flts(u,h[,past])
```

Ex :

```
// criando o polinômio do(filtro)
[h,hm,fr]=wfir('lp',33,[.2 0],'hm',[0 0]);
// criando o sinal a ser filtrados
t=1:200;
x1=sin(2*%pi*t/20);
x2=sin(2*%pi*t/3);
x=x1+x2;
// exibindo o sinal
plot2d(x);
//criando a função de transferência
z=poly(0,'z');
hz=syslin('d',poly(h,'z','c')./z**33);
//filtrando o sinal
yhz=flts(x,hz);
//exibindo o resultado
clf();
plot2d(yhz);
```

## 7. Diagrama de Bode de um sistema

A curva de resposta em frequência de um sistema pode ser gerada através da função Bode, com a seguinte sintaxe :

```
bode(h,fmin,fmax[,step][,comments])
```

onde h é a função de transferência a ser avaliada, fmin e fmax são os limites em frequência e step (parâmetro opcional) e o intervalo entre pontos.

Ex1:

```
// criando a função de transferência
s=poly(0,'s');
h1=1/real((s+2*%pi*(15+100*%i))*(s+2*%pi*(15-100*%i)))
h1=syslin('c',h1);
// plotando o diagrama de bode
bode(h1);
```

Ex2 – dois sistemas em série

```
// definindo sistema 1 pela notação polinomial
s=poly(0,'s');
h1=1/real((s+2*%pi*(15+100*%i))*(s+2*%pi*(15-100*%i)));
h1=syslin('c',h1);
// definindo sistema 1 pela notação espaço de estados
a=-2*%pi;b=1;c=18*%pi;d=1;
sl=syslin('c',a,b,c,d);
// convertendo para notação polinomial
h2=ss2tf(sl)
// conectando em série
h= h1*h2
//traçando o diagrama de bode
bode(h);
```

## 8. Diagrama de Pólos e Zeros

Outra ferramenta importante na análise de sistemas, sejam eles discretos ou contínuos, é o diagrama de pólos e zeros.

Para a visualização dos pólos e zeros de uma função de transferência no plano Z, utiliza-se a função  $plrz(h)$ ,

Ex:

```
// criando um polinômio função de transferência:  
hz=iir(4,'lp','butt',[.25 0],[0 0])  
// calculando os zeros:  
zhz = roots(numer(hz))  
// calculando os polos  
phz=roots(denom(hz))  
//exibindo o diagrama de pólos e zeros  
plzr(hz)
```

Exercícios :

10. Tomando como base o exemplo do item 6 , determine :

- a resposta espectral (fft) do sinal antes de ser filtrado
- a resposta temporal do sinal filtrado
- o diagrama de bode da função de transferência (hz)
- o diagrama de pólos e zeros da função de transferência
- Qual é o efeito do filtro no sinal ? Como poderias justificar a sua resposta.

11. Analise os exemplos do item 7 (diagrama de bode) e explique qual seria o efeito destes filtros (h, h1 e h2) em um sinal. Elabore um exemplo de aplicação , como feito no exercício 10.

12. Trace o diagrama de pólos e zeros das funções de transferência dos exemplos do item 7 (h, h1 e h2)

13. Trace o diagrama de bode do exemplo do item 8.

14. Qual é a relação entre os pólos e zeros de uma função de transferência e a sua resposta em freqüência ?